

HP42S VERTICAL ALIGNMENT

D'Zign

NOTICE

No express or implied warranty is made by D'Zign Land Survey & Development or the author with regard to the procedures and program material offered or their merchantability or their fitness for any particular purpose. The keystroke procedures and program material are made available solely on an "as-is" basis, and the entire risk as to their quality and performance is with the user. Should the procedures or program material prove defective, the user (and not D'Zign Land Survey & Development nor any other party) shall bear any and all cost of all necessary correction and all incidental or consequential damages. D'Zign Land Survey & Development and/or the author shall not be liable for any incidental or consequential damages in connection with or arising out of the furnishing, use, or performance of the keystroke procedures or program material.

HP42S VERTICAL ALIGNMENT

Library of Congress Catalog Card No. 89-81219
ISBN 0-944889-07-7

Manufactured in the United States of America

2nd printing August 1990

published by D'Zign Land Survey & Development
Glendale, California, U.S.A. 1989

ISBN 0-944889-07-7

TECHNICAL ASSISTANCE

The program material, instructions and procedures contained in this book assume that the user has a working knowledge of both surveying and the general operation of the HP-42S calculator.

Technical assistance is limited to verification of the results shown in the various examples used in the book.

If you have any questions or suggestions regarding this book or other **D'Zign** publications, please feel free to call us. The number is (818) 507-7408, and someone is available to answer technical questions from 3:00 A.M. to 5:30 A.M. and from 4:30 P.M to 6:30 P.M. (Pacific time), as a service to users from other time zones.

Hewlett-Packard has produced a really powerful calculator at a very good price, the HP-42 Scientific Calculator, which lends itself nicely to solving surveying problems. It can not be programmed by insertion of a module, or with a card reader, like the HP-41 series, but it has a really simple system for typing in a program.

the operations index

To find a function for the first time, HP has provided an "Operations Index" on pages 310 through 335 of the instruction manual, which tells you exactly what keystrokes to use to type in the function you want.

Even better, this index gives you the page number that you can refer to if you want to know more about the function you are using. If, while typing in a program, you aren't sure how to input a particular function, simply refer to the Operations Index.

the softkey menus

All of the programs in this booklet take advantage of the "softkey" menu system built into this calculator. When you want to start a program you stroke **XEQ** and then the softkey corresponding to the program you want, from the menu displayed in the bottom half of the screen.

the programs

The purpose of this booklet is to provide the user with the best, and most flexible system of programs for solving vertical alignment problems possible.

Programs are included for solution of symmetrical and asymmetrical parabolic curves and grade intersection solutions. We have also included routines which will let you store a string of vertical intersection points and then use them as needed for calculation of the curves. The amount of programming which you input depends on your individual needs.

Because some of these are long programs, we suggest that you approach the programming in stages. You are less apt to have programming errors if you don't try to do the whole job at one sitting.

This booklet contains a short routine which can calculate the station and elevation for the intersection of any two grade lines which pass through known points. Once the intersection location is known, minor adjustments are usually made to put the intersection at an even (or at least reasonable) station.

There are keystroke examples included in the text, to allow step-by-step practice with the program.

The use of a printer is not required, but does make the whole job easier, Hewlett-Packard has the InfraRed Printer available for the 42S, and one feature of this calculator is that it already has the InfraRed transmitter built in.

subroutines

Because of the way the calculator works, we will start by input of some subroutines.

Once the subroutine has been input, its name appears in the menu when you stroke XEQ, and all you have to do to add it as a step in the program you are typing in is stroke GTO or XEQ followed by keystroking the key corresponding to the subroutine to input the program step GTO XXX or XEQ XXX.

```
00 { 113-Byte Prgm }
01▶LBL "YN"
02 "YES"
03 KEY 1 GTO 01
04 "NO"
05 KEY 2 GTO 02
06 MENU
07 STOP
08▶LBL 01
09 SF 10
10 GTO 03
11▶LBL 02
12 CF 10
13▶LBL 03
14 CLMENU
15 EXITALL
16 RTN
```

getting started

Begin by stroking the shift key, then the **XEQ** key. The display will show a menu which will be blank (if you haven't yet input any programs) except for **.END.** on the left. The keys just below each of the menu portions will correspond to the menu instruction above it. Stroke the key just below the **.END.** in the display.

```
00▶{ 0-Byte Prgm }
01 .END.
```

Next, go into **program mode** by stroking the shifted R/S key, and you should have a display similar to the one shown to the left. Begin typing in the program "**YN**" from the last page.

quick tip

Program steps 03 and 05 use a function which stores the prompt to the menu, and at the same time assigns the key.

To access the function, stroke ▢ PGM.FCN ▲ [illegible]

You'll receive a prompt, **KEY _**. Stroke the key number (we'll use 1 as the example), and you will get a prompt, **KEY 1 GTO _ _**.

In this case (step 03) answer 01 to complete the program step.

next subroutine

This one has three steps we might want to review before you start:

STO ST Y — To use a store or recall function involving the stack, stroke STO (or RCL) . This brings up a menu from which you can select the function to complete the program step.

1E2 — Third row down, second from the right, is a key, E. Stroke this key, then the number (in this case 2). Nothing will happen until you input the next step, usually 'times' or 'divide'.

⊢"↳Elev. " — The ⊢ symbol is "append", which adds to what is already in the **alpha** register. The symbol, ↳, is "line feed", and we use it to control the display. You can input it by stroking

▢ ALPHA ▼ [illegible] ▼ [illegible]

The program shown below may be typed in as a continuation of the previous one. Just keep adding steps to your program.

This program, "**PST**" is a subroutine that outputs the station and elevation during calculations.

```
17▶LBL "PST"
17▶LBL "PST"  26 IP          35 10        43 RTN
18 CF 29      27 FC? 20      36 X>Y?      44 FS? 11
19 FIX 00     28 " Sta. "    37 ⊢"0"      45 ⊢"↳ Elev "
20 FC? 11     29 ARCL ST X   38 ARCL ST Y  46 FC? 11
21 CLA        30 -           39 R↑        47 ⊢"   "
22 STO ST Y   31 ⊢"+"        40 R↑        48 ARCL 09
23 1E2        32 FIX 03      41 SF 29     49 FC? 11
24 ÷          33 1E2         42 FS? 19    50 ⊢"↳"
25 ENTER      34 x                        51 AVIEW
```

After typing in the program as shown, stroke XEQ, enter **alpha mode**, and type in "END". When you leave alpha, you should now have step 51 END as a last program step.

There are three more short subroutines to input, below. We can group them where we want them in program memory by stroking ▢ XEQ [illegible] and then scrolling up one step (to 00) by stroking ▲.

```
01▶LBL "CL"          09▶LBL "FCL"       16▶LBL "ZIP"
02 ΣREG 00           10 0.013           17 CLA
03 CLΣ               11▶LBL "FN"        18 CF 21
04 ΣREG 11           12 CF IND ST X     19 "↳ WON'T WORK!"
05 CLΣ               13 ISG ST X        20 AVIEW
06 0                 14 GTO "FN"        21 STOP
07 STO 24            15 RTN             22 GTO "PVI"
08 RTN                                  23 END
```

The next one is longer. This time go to the top of "**CL**" and scroll up one to begin.

```
00 { 279-Byte Prgm }
01▶LBL "PVI"
02 SF 21
03 CF 19
04 SF 11
05 CLA
06 "↳ 1st. Sta. ↑ E"
07 ⊢"lev."
08 PROMPT
09 CLA
10 "↳ 2nd Sta. ↑ E"
11 ⊢"lev."
12 PROMPT
13 STO 04
14 R↓
15 STO 02
16 X<>Y
17 STO 03
18 R↓
19 X<>Y
20 STO 01
21 -
22 STO 05
23 R↓
24 -
25 CLA
26 "↳ 1st. % Grade?"
27 PROMPT
28 "↳ 2nd % Grade?"
29 PROMPT
30 1E2
31 ÷
32 STO 06
33 X<>Y
34 1E2
35 ÷
36 STO 07
37 R↓
38 ÷
39 RCL+ 05
40 RCL× 06
41 RCL 07
42 RCL- 06
43 ÷
44 +/-
45 RCL+ 01
46 STO 23
47 RCL 01
48 X>Y?
49 XEQ "ZIP"
50 CLX
51 RCL 02
52 X≤Y?
53 XEQ "ZIP"
54 R↓
55 RCL 23
56 RCL- 01
57 RCL× 07
58 RCL+ 03
59 STO 22
60 FC? 55
61 GTO 00
62 RCL 03
63 STO 09
64 RCL 01
65 XEQ "PST"
66 ADV
67 RCL 07
68 1E2
69 ×
70 CLA
71 "↳Grade 1 = "
72 ARCL ST X
73 ⊢"%"
74 AVIEW
75 ADV
76▶LBL 00
77 RCL 22
78 STO 09
79 RCL 23
80 FC? 55
81 CF 21
82 CLA
83 "P.V.I. is at"
84 AVIEW
85 PSE
86 SF 21
87 XEQ "PST"
88 ADV
89 FC? 55
90 RTN
91 RCL 06
92 1E2
93 ×
94 CLA
95 "↳Grade 2 = "
96 ARCL ST X
97 ⊢"%"
98 AVIEW
99 ADV
100 RCL 04
101 STO 09
102 RCL 02
103 XEQ "PST"
104 ADV
```

Finish by adding END to the program. What you have just typed in is a program which will solve the intersection of two grade lines which pass through known points of station and elevation.

user instructions

Start by stroking XEQ [redacted]

prompt: 1st. Sta. ↑ Elev.

Input the first (down-station) station, and stroke ENTER

Input the elevation for that station and stroke R/S

prompt: 2nd Sta. ↑ Elev.

Input the second (up-station) station, stroke ENTER

Input the elevation for that station and stroke R/S

prompt: 1st. % Grade?

Input the percent of grade for the line through the first station. If the grade is negative, stroke +/- R/S

prompt: 2nd % Grade?

Input the outgoing grade (percent). If negative, +/- R/S

The output will depend on whether or not you are using a printer. If you are, the output is the first station,

first grade, "P.V.I. is at" will show briefly on the display, then the solution station and elevation will be displayed. This is followed by output of the second grade and the ending station and elevation.

Without the printer you will see the "P.V.I. is at" in the display, and then the program will stop with the intersection station and elevation in the display.

error message

If you give the calculator an impossible set of instructions, WON'T WORK! will appear in the display. This is the equivalent of "data error". If the problem is that you may have stroked a wrong key during input, stroke R/S to start over.

The amount of time that "P.V.I. is at" will be seen in the display is about one second. If you want it to show longer you can add additional pauses (PSE) after step 84 in the program. If you want to see less of it, delete step 85 (or even steps 80 through 86, and you won't see it at all).

keystroke examples

As examples of the keystrokes used for "**PVI**", we can use the information shown in the two illustrations to the right.



In the first example we have a +2% grade in, passing throuth station 10+00 at elevation 120.0, and a -1.6% grade out, passing through station 16+00 at elevation 125.0.

The keystrokes are as shown on the next page.

example one

keystroke:

XEQ [illegible]

prompt: 1st. Sta. ↑ Elev.

keystrokes:

1 0 0 0 ENTER

1 2 0 R/S

prompt: 2nd Sta. ↑ Elev.

keystrokes:

1 6 0 0 ENTER

1 2 5 R/S

prompt: 1st. % Grade?

keystrokes:

2 R/S

prompt: 2nd % Grade?

keystrokes:

1 . 6 +/- R/S

output:

```
Sta. 10+00.000
Elev 120.000

Grade 1 = 2.000%

P.V.I. is at
 Sta. 14+05.556
 Elev 128.111

Grade 2 = -1.600%

 Sta. 16+00.000
 Elev 125.000
```

example two

keystrokes:

XEQ [illegible]

prompt: 1st. Sta. ↑ Elev.

keystrokes:

1 0 0 0 ENTER

1 2 0 R/S

prompt: 2nd Sta. ↑ Elev.

keystrokes:

1 6 0 0 ENTER

1 0 4 . 5 R/S

prompt: 1st. % Grade?

keystrokes:

2 . 5 +/- ENTER

prompt: 2nd % Grade?

keystrokes:

3 +/- R/S

output:

```
Sta. 10+00.000
Elev 120.000

Grade 1 = -2.500%

P.V.I. is at
 Sta. 15+00.000
 Elev 107.500

Grade 2 = -3.000%

 Sta. 16+00.000
 Elev 104.500
```

The main program for vertical alignment solves grade or vertical curve problems, calculating the elevations at any station along the vertical alignment. You can also solve for the station when the elevation is known, on the tangents and in **symmetrical** curves.

types of verticals

Vertical curves are usually described as 'crest' or 'sag' verticals, as shown to the right.



The program will also calculate the turning point (high point on crest, low point on sags) station and elevation. When both tangent grades have the same sign, there is no actual turning point, and the program will tell you that there is no high or low point in the curve.

The symbol, **e**, is commonly used to denote the distance from the tangent intersection to the curve. In this program, the value is computed, and shown as the second (POVC) elevation in the curve data.

A symmetricl curve is one in which the lengths of the two tangents are equal. The program allows selection by input of the length, **L**, if it is a symmetrical curve, or lengths **L1** and **L2** if it is not.

intervals

Automatic calculation of the elevations at a given interval of stationing is possible. The interval instruction is cancelled at the end of each tangent or curve, and must be reset for the next segment. This allows the use of a different interval along the tangent than in the curve.

To allow you to take advantage of the interval option, but still be able to calculate specific stations within the

segment you are calculating, at the end of each tangent or curve you are prompted for additional station input.

the program

To begin typing in the program steps, stroke [ ] XEQ and then the key under ".**END.**". Your display, when you enter program mode, should be as shown to the right.

```
00 { 0-Byte Prgm }
01▶.END.
```

Scroll up once and type in the program below. It may be better to not try to do it all at one sitting. When you leave program mode and turn the calculator off, the program pointer will still be at the point where you left off when you turn it back on.

```
00 { 1174-Byte Prgm }
01▶LBL "YA"
02 XEQ "CL"
03 XEQ "FCL"
04 CF 19
05 SF 20
06 SF 21
07 SF 01
08 XEQ 00
09▶LBL 03
10 FS? 11
11 XEQ 01
12▶LBL 02
13 XEQ 28
14 CLA
15 FS? 03
16 PROMPT
17 FC? 00
18 PROMPT
19▶LBL 29
20 FS? 00
21 XEQ 04
22 STO 07
23 RCL 12
24 X≤Y?
25 XEQ 05
26 R↓
27 RCL- 00
28 RCL× 02
29 RCL+ 01
30 STO 09
31 RCL 07
32 XEQ "PST"
33 FS?C 03
34 GTO 32
35 GTO 03
36 RTN
37▶LBL I
38 STO 04
39 RCL 07
40 SF 00
41 FS? 01
42 GTO 29
43 FC? 01
44 GTO 08
45 RTN
```

```
46▶LBL 00
47 "↳Station?"
48 PROMPT
49 STO 00
50 "↳Elevation?"
51 PROMPT
52 STO 01
53 "↳% Grade?"
54 PROMPT
55 1E2
56 ÷
57 STO 02
58▶LBL 01
59 CF 11
60 XEQ 15
61 "NEXT"
62 PROMPT
63▶LBL 24
64 XEQ 16
65▶LBL 23
66 FS? 81
67 XEQ 20
68 FS? 82
69 XEQ 21
70 FS? 83
71 XEQ 22
72 RCL 15
73 RCL- 00
74 RCL× 02
75 RCL+ 01
76 STO 16
77 GTO 02
78 RTN
79▶LBL 19
80 STO 05
81 2
82 ÷
83 STO 17
84 STO 18
85 SF 08
86 GTO 23
87▶LBL 17
88 STO 17
89 STOP
90▶LBL 18
91 STO 18
92 +
93 STO 05
94 SF 08
95 GTO 23
96▶LBL 25
97 STO 12
98 SF 81
99 GTO 24
100▶LBL 26
101 STO 15
102 SF 82
103 GTO 24
104▶LBL 27
105 STO 10
106 SF 83
107 GTO 24
108▶LBL 20
109 CF 81
110 RCL 12
111 RCL+ 17
112 STO 15
113 RCL+ 18
114 STO 10
115 RTN
116▶LBL 21
117 CF 82
118 RCL 15
119 RCL- 17
120 STO 12
121 RCL+ 05
122 STO 10
123 RTN
124▶LBL 22
125 CF 83
126 RCL 10
127 RCL- 05
128 STO 12
129 RCL+ 17
130 STO 15
131 RTN
132▶LBL 04
133 RCL+ 04
134 RTN
135▶LBL 32
136 XEQ 35
137 FC? 10
138 GTO 29
139 SF 03
140 GTO 03
141▶LBL 36
142 RCL 15
143 X<>Y
144 X<Y?
145 SF 06
146 X>Y?
147 CF 06
148 RTN
149▶LBL 34
150 XEQ 35
```

```
151 FC? 10
152 GTO 12
153 SF 03
154 CF 06
155 XEQ 08
156▶LBL 35
157 CLA
158 CLMENU
159 CF 21
160 "Additional Stat"
161 ⊢"ion?"
162 AVIEW
163 SF 21
164 XEQ "YN"
165 RTN
166▶LBL 05
167 CLMENU
168 FS?C 00
169 XEQ 32
170 FS? 04
171 STOP
172 CF 01
173 ADV
174 RCL 12
175 RCL- 00
176 RCL× 02
177 RCL+ 01
178 STO 01
179 STO 09
180 RCL 12
181 STO 00
182 SF 11
183 CLA
184 "BVC "
185 XEQ 07
186 CF 29
187 FIX 00
188 "   "
189 ARCL 05
190 ⊢" Ft V.C.↳"
191 AVIEW
192 FIX 03
193 "↳Grade In = "
194 RCL 02
195 1E2
196 ×
197 ARCL ST X
198 ⊢"%"
199 AVIEW
200 SF 29
201 CLA
202 RCL 16
203 STO 09
204 RCL 15
205 CLA
206 ⊢" PVI "
207 XEQ 07
208 "Grade Out?↳"
209 PROMPT
210 FS? 19
211 ADV
212 FS? 19
213 GTO 16
214 1E2
215 ÷
216 STO 03
217 SIGN
218 RCL 02
219 SIGN
220 X=Y?
221 SF 02
222 RCL 03
223 RCL- 02
224 2
225 ÷
226 RCL÷ 05
227 STO 14
228 RCL× 18
229 RCL÷ 17
230 STO 06
231 RCL 18
232 RCL× 03
233 RCL+ 16
234 STO 11
235 RCL 14
236 RCL 17
237 RCL× 18
238 ×
239 RCL 17
240 RCL× 02
241 RCL+ 01
242 +
243 CLA
244 STO 09
245 "↳POVC "
246 ARCL 09
247 AVIEW
248 ADV
249 RCL 14
250 RCL× 17
251 RCL÷ 18
252 STO 08
253 RCL 02
254 RCL 06
255 2
```

```
256 ×
257 ÷
258 +/-
259 STO 13
260 RCL 02
261 X↑2
262 RCL 06
263 4
264 ×
265 ÷
266 RCL 01
267 X<>Y
268 -
269 STO 09
270 FS? 08
271 XEQ 09
272 RCL 00
273 RCL+ 13
274 CLA
275 FC? 02
276 ⊢" G=0% "
277 FC? 02
278 XEQ 07
279 FS?C 02
280 XEQ 33
281 FIX 03
282 "Grade Out = "
283 RCL 03
284 1E2
285 ×
286 ARCL ST X
287 ⊢"%↳"
288 AVIEW
289 CF 11
290 RCL 00
291 STO 07
292 ADV
293 GTO 08
294 RTN
295▶LBL 33
296 CLA
297 "No Hi/Low Point"
298 ⊢"↳"
299 AVIEW
300 CLA
301 RTN
302▶LBL 10
303 FS? 08
304 R↓
305 FS? 06
306 GTO 12
307 SF 06
308 FC? 08
309 RTN
310 FS? 08
311 SF 11
312 FS? 08
313 GTO 11
314 RTN
315▶LBL 12
316 FS?C 00
317 XEQ 34
318 CF 08
319 CF 06
320 RCL 03
321 STO 02
322 RCL 11
323 STO 01
324 STO 09
325 RCL 10
326 STO 00
327 STO 07
328 CLA
329 "EVC "
330 XEQ 07
331 SF 01
332 SF 11
333 GTO 03
334 RTN
335▶LBL 13
336 1
337 +
338 STO 12
339 SF 01
340 SF 04
341 GTO 03
342 RTN
343▶LBL 11
344 FS? 11
345 GTO 14
346 XEQ 28
347 CLA
348 FS? 03
349 GTO 14
350 FC? 00
351 PROMPT
352 FS? 00
353 XEQ 04
354▶LBL 14
355 CF 11
356 STO 07
357 RCL 10
358 X≤Y?
359 GTO 12
360 -
```

```
361 STO 13
362 X↑2
363 RCL× 08
364 RCL 13
365 RCL× 03
366 +
367 RCL+ 11
368 STO 09
369 RCL 07
370 XEQ "PST"
371 FS? 03
372 GTO 34
373 GTO 11
374 RTN
375▶LBL 09
376 RCL 17
377 RCL 13
378 X≤Y?
379 RTN
380 RCL 03
381 RCL 08
382 2
383 ×
384 ÷
385 +/-
386 STO 13
387 RCL 03
388 X↑2
389 RCL 08
390 4
391 ×
392 ÷
393 RCL 11
394 X<>Y
395 -
396 STO 09
397 RCL 05
398 STO+ 13
399 RTN
400▶LBL 08
401 XEQ 28
402 CLA
403 FS? 03
404 PROMPT
405 FC? 00
406 PROMPT
407▶LBL 31
408 CF 11
409 FS? 00
410 XEQ 04
411 STO 07
412 FS? 03
413 XEQ 36
414 FC? 06
415 RCL 15
416 FS? 06
417 RCL 10
418 X≤Y?
419 XEQ 10
420 R↓
421 RCL- 00
422 STO 13
423 X↑2
424 RCL× 06
425 RCL 13
426 RCL× 02
427 +
428 RCL+ 01
429 STO 09
430 RCL 07
431 XEQ "PST"
432 FS?C 03
433 GTO 34
434 GTO 08
435▶LBL 07
436 SF 20
437 SF 11
438 XEQ "PST"
439 CF 11
440 CLA
441 ADV
442 RTN
443▶LBL E
444 STO 09
445 RCL- 01
446 FS? 01
447 XEQ A
448 CF 11
449 CF 19
450 FS? 01
451 GTO 02
452 +/-
453 RCL 06
454 RCL 02
455 R↓
456 ÷
457 R↑
458 LASTX
459 ÷
460 2
461 ÷
462 +/-
463 ENTER
464 ENTER
465 X↑2
```

```
466 R↑
467 -
468 SQRT
469 -
470 LASTX
471 R↑
472 +
473 RCL+ 00
474 STO 19
475 X<>Y
476 RCL+ 00
477 STO 07
478 SF 19
479 SF 11
480 XEQ b
481 RCL 19
482 STO 07
483 "and "
484 XEQ "PST"
485 AVIEW
486 CLA
487 CF 11
488 CF 19
489 ADV
490 GTO 08
491 RTN
492▶LBL A
493 RCL÷ 02
494 RCL+ 00
495 STO 07
496▶LBL b
497 "Elev. "
498 ARCL 09
499 ⊢" at"
500 AVIEW
501 SF 11
502 SF 19
503 "Sta "
504 XEQ "PST"
505 AVIEW
506 FC? 01
507 RTN
508 CF 11
509 ADV
510 GTO 28
511 RTN
512▶LBL 15
513 CLMENU
514 "B.V.C"
515 KEY 1 GTO 25
516 "P.V.I"
517 KEY 2 GTO 26
518 "E.V.C"
519 KEY 3 GTO 27
520 "END@"
521 KEY 6 GTO 13
522 MENU
523 RTN
524▶LBL 28
525 CLA
526 CLMENU
527 "STA"
528 KEY 1 GTO 30
529 "ELEV"
530 KEY 2 GTO E
531 "INTVL"
532 KEY 3 GTO I
533 MENU
534 RTN
535▶LBL 16
536 CLMENU
537 "L"
538 KEY 1 GTO 19
539 "L 1"
540 KEY 2 GTO 17
541 "L 2"
542 KEY 3 GTO 18
543 MENU
544 STOP
545▶LBL 30
546 FS? 01
547 GTO 29
548 FC? 01
549 GTO 31
```

If you scroll down once, you will see the permanent end, **.END.**, still in place. If you decide to add another program later, you will need to add an end to this program before starting.

user instructions

Execute "**VA**" to bring up the first prompt.

prompt: Station?

Input any station which has a known elevation and stroke

R/S

prompt: Elevation?

Input the elevation of the station, and then stroke

R/S

prompt: % Grade?

Input the grade (percent). If the grade is negative, stroke +/-. Stroke

R/S

prompt: NEXT

At this point you may tell the calculator to stop at a given station (when you are using the automatic interval, it will stop calculation at that station), or you can give it one of the indicated stations along the vertical curve.

To stop calculation at a given station, input the station and stroke

or

Input the station at the desired part of the curve, stroke the appropriate key,

prompt:

If the curve is symmetrical, input the total length of the curve and stroke

or

If the curve is asymmetrical, input the length of the first tangent, stroke

Input the length of the second tangent and stroke

prompt: 

options

If you want to calculate the station at which a given elevation occurs, input the elevation and stroke

The **output** will be the station at which the elevation occurs.

```
Elev. 125.440 at
Sta 12+72.000
```

prompt: 

Input the first station for which you want to calculate an elevation. Stroke

The **output** will be the station and the elevation.

```
10+00.000    120.000
```

After output of the initial station, you can set the interval if you like.

Input the interval you would like to use and stroke

Note: The interval-spaced calculations will begin at the last station output. If you calculated an elevation (above) you will want to output a starting station before input of the interval.

The program will continue to calculate the elevations until the next BVC station is reached. If **END@** was selected, it will stop at the designated station. If you are not using a printer, continue stroking R/S as the solutions appear.

When a BVC station is reached, the program will automatically output the curve data through the PVI station and elevation. Continue stroking R/S for the output, if you are not using a printer.

prompt: Grade Out?

Input the grade (percent). If negative, +/- , and stroke R/S

the **output** will be the remainder of the curve data, including the turning point and the elevation on the curve at the PVI station (POVC).

At this point you can calculate (symmetrical curve) the station at which a given elevation occurs, if needed.

When the end of the curve is reached, the calculator will prompt for additional (oddball) stations:

prompt: Additional Station?

Answer yes, if you want to calculate a station that was not one of the ones at regular interval, by stroking

or

Answer no if you do not need additional stations. This prompt will continue after each output until you answer "no", Then the EVC information will be output.

keystroke examples

In the first example on page 8, the PVI was calculated to be at station 14+05.556. A curve using this PVI could still begin and end at even stations. Let's use 13+00 and 15+00, for a 200' vertical curve, by calculating it as an asymmetrical curve with L1 = 105.556 and L2 = 94.444.

We can use this as a first keystroke example, and assume that we want to calculate elevations at 50' intervals along the tangent, and 25' intervals through the curve.

prompt: Station?

keystrokes:

1 0 0 0 R/S

prompt: Elevation?

keystrokes:

1 2 0 R/S

prompt: % Grade?

keystrokes:

2 R/S

prompt: NEXT

keystrokes:

1 3 0 0

prompt:

keystrokes:

1 0 5 . 5 5 6

9 4 . 4 4 4

prompt:

keystrokes:

1 0 0 0

output:

10+00.000 120.000

prompt:

keystrokes:

5 0

output:

10+50.000	121.000
11+00.000	122.000
11+50.000	123.000
12+00.000	124.000
12+50.000	125.000

prompt: Additional Station?

keystrokes:

output:

BVC 13+00.000
Elev 126.000

200 Ft V.C.
Grade in = 2.000%
PVI 14+05.556
Elev 128.111

prompt: Grade Out?

keystrokes:

1 . 6 +/- R/S

output:

POVC 127.214

G=0% 14+20.469
Elev 127.236

Grade Out = -1.600%

prompt:

keystrokes:

2 5

output:

13+25.000	126.450
13+50.000	126.799
13+75.000	127.047
14+00.000	127.195
14+25.000	127.234
14+50.000	127.149
14+75.000	126.937

prompt: Additional Station?

keystrokes:

output:

EVC 15+00.000
Elev 126.600

The curve below is symmetrical, and we'll use it as a second example, at the same time trying out the **ELEV** option. Assume that we want to know where elevation 108.00 will be in the curve, and where elevation 110.00 will be in the tangent following the curve.



Enter the program by executing "**VA**".

prompt: Station?

keystrokes:

1 0 5 0 R/S

prompt: Elevation?

keystrokes:

1 0 6 R/S

prompt: % Grade?

keystrokes:

1 . 5 +/- R/S

prompt: NEXT

keystrokes:

1 0 5 0 ENTER

prompt:

keystrokes:

2 0 0

prompt:

keystrokes:

1 0 5 0

output:

```
BVC 10+50.000
 Elev 106.000

   200 Ft V.C.
Grade in = -1.500%
 PVI 11+50.000
 Elev 104.500
```

prompt: Grade Out?

keystrokes:

4 R/S

output:

```
POVC 105.875
 G=0% 11+04.545
 Elev 105.591
Grade Out = 4.000%
```

prompt:

keystrokes:

1 0 8

output:

```
ELEV 108.000 at
Sta 9+72.180
and 12+36.911
```

prompt: [Sta] [ELEV] [CURVE]

keystrokes:

[1] [1] [0] [0] [Sta]

output:

11+00.000 105.594

prompt: [Sta] [ELEV] [CURVE]

keystrokes:

[5] [0] [INTRVL]

output:

11+50.000 105.875

12+00.000 106.844

prompt: Additional Station?
[YES] [NO]

keystrokes:

[NO]

output:

EVC 12+50.000
Elev 108.500

prompt: NEXT
[ALIGN] [ZOOM] [DATA] [EDIT]

keystrokes:

[1] [6] [0] [0] [ENTER]

prompt: [Sta] [ELEV] [CURVE]

keystrokes:

[1] [1] [0] [ELEV]

output:

ELEV 110.000 at
Sta 12+87.500

prompt: [Sta] [ELEV] [CURVE]

keystrokes:

[1] [3] [0] [0] [Sta]

output:

13+00.000 110.500

prompt: [Sta] [ELEV] [CURVE]

keystrokes:

[5] [0] [INTRVL]

output:

13+50.000 112.500

14+00.000 114.500

14+50.000 116.500

15+00.000 118.500

15+50.000 120.500

16+00.000 122.500

Note that the **ELEV** option, used in a curve, will output two points in most cases. They may NOT be in the curve and, if they occur on the tangents, are not correct for the tangent. Visual inspection of the output will tell you which ones to use.

Also, remember that if you have used the elevation option and want to use the interval option after that, you need to input a starting station first.



MORE HP42S PROGRAMS

Spiral Curves (booklet, $9.50)

Calculate the coordinates to any station, or offset to a station, within a spiral system. Options include coordinate output, auto-inverse, or both.

Calculates intersections of the entrance or exit spiral with a circular curve or straight line.

Triangle Solutions (booklet, $8.00)

The 42S version of the most complete triangle solutions program ever available. Solves with any of the following knowns: ASA SAA SAS SSA SSS Area-SS Area-AA Area-SA.

Topography (booklet, $8.00)

This one turns your 42S into a manual data collector, complete with a labeling system that you can customize to suit the type of topo work you do.

All shots are stored as finished data, by shot number, for later output. Choice of 3-D coordinates or Station-Offset-Elevation for the output.

EDM Slope Staking (booklet, $8.00)

Set up anywhere near an alignment and slope stake it. Sets slope stakes from the remote instrument location directly. Includes a three-point resection program for finding the instrument's location by either station-offset or coordinates. All data needed to mark the stake is output (or may be stored), and there is a subroutine for setting the reference stake.

Urban Survey Programs (book, $20.00)

Contains programs for stakeout (or design) of street intersections, returns, culs-de-sac, bulbs and knuckles. A special program calculates and prorates blocks from the street intersections (with auto-store and auto-inverse of the points) from the street intersection setups for A.L.T.A. surveys.

Also contains layout programs for curb and gutter, or storm/sanitary sewers, which can store the data and later print out cut sheets before leaving the job site.

$8.00

ISBN 0-944889-07-7